

Satisfiability III (sat3)

— Run LPL Code , HTML Document —

Problem: Is the following set of statements correct? “If Superman were able and willing to prevent evil, he would do so. If Superman were unable to prevent evil, he would be impotent and if he were unwilling to prevent evil, he would be malevolent. Superman does not prevent evil. If Superman exists, he is neither impotent nor malevolent. Therefore, Superman does not exist.” (see [1]).

Modeling Steps

1. We introduce 6 Boolean propositions as follows:

```
a "Superman is able to prevent evil"
w "Superman is willing to prevent evil"
i "Superman is impotent"
m "Superman is malevolent"
p "Superman prevents evil"
e "Superman exists"
```

2. Then the first statement can be formulated as: $a \wedge w \rightarrow p$.
3. The second is: $(\bar{a} \rightarrow i) \wedge (\bar{w} \rightarrow m)$.
4. The third is: \bar{p} .
5. The fourth is: $e \rightarrow \overline{i \vee m}$.
6. The last statement (the consequence) is: \bar{e} .

Hence, we need to show the following Boolean expression:

$$[(a \wedge w \rightarrow p) \wedge (\bar{a} \rightarrow i) \wedge (\bar{w} \rightarrow m) \wedge \bar{p} \wedge (e \rightarrow \overline{i \vee m})] \rightarrow (\bar{e})$$

Listing 1: The Complete Model implemented in LPL [3]

```
model SAT3 "Satisfiability III";
  binary variable
    a "Superman is able to prevent evil";
    w "Superman is willing to prevent evil";
    i "Superman is impotent";
    m "Superman is malevolent";
    p "Superman prevents evil";
    e "Superman exists";
  constraint
    s1: a and w -> p;
    s2: (~a -> i) and (~w -> m);
    s3: ~p;
    s4: e -> i nor m;
  minimize ne: ~e "Superman does not exist";
  Write('%s', if (e, 'Superman_exists.', 'Superman_does_not_exist'));
end
```

Solution: The Boolean statement is correct, that is, the argumentation is correct and unfortunately, Superman does not exist.

Questions

1. Verify that LPL's translation is correct, by generating the CNF using the method given in model [sat1](#)¹
2. Verify that the argument *cannot* be proven with the LP relaxation.

Answers

1. The LPL translation is as follows

```
min:    -e;
s1:     p - w - a >= -1;
s2:     i + a >= 1;
s4:     -i - e >= -1;
X56X:   m + w >= 1;
X57X:   -m - e >= -1;
```

2. Build the LP relaxation of the problem and solve it. The optimum is -1 . This is smaller than the value of $B(k)$ which is 0 . Therefore we have no proof.

References

- [1] D. Gries and F.B. Schneider. *A Logical Approach to Discrete Math*. Springer, p 37, 1994.
- [2] MatMod. Homepage for Learning Mathematical Modeling : <https://matmod.ch>.
- [3] Hürlimann T. Reference Manual for the LPL Modeling Language, most recent version. <https://matmod.ch/lpl/doc/manual.pdf>.

¹<https://lpl.matmod.ch/lpl/Solver.jsp?name=/sat1>